Example 1 (1-sample T-Test): The data file 'Score with 1-Sample T-Test Calculations.xlsx. has final scores for the STATS 101 class at a university. Test if the true mean μ for STATS 101 class equals 90.

This problem is formulated as testing H_0: μ = 90 vs. H_1: μ ≠ 90.

Start with SCORE data in cells A2:A61. Microsoft Excel does not provide a 1-sample t-test /confidence interval function, so you need to compute these in Excel. The Score data file has all of these calculations.

Running the 1-sample t-test

Calculate the mean (Figure 1a), sd (Figure 1b), the t-statistic (Figure 1c) in Excel from the following formula:

$$t_{OBS} = \frac{\bar{x} - \mu_0}{s/\sqrt{n}} = \frac{\bar{x} - 90}{s/\sqrt{n}}$$

and then using the Excel function tdist to compute the P-value using the formula $\text{P-value} = 2 \times P(t_{n-1} > |t_{OBS}|)$ as shown in Figure 1d.

Note that '=tdist(|t_{OBS}|,n-1,2) gives the P-value for 2-sided alternative (Figure 1d), and

'=tdist(|t_{OBS}|,n-1,1) gives the P-value for 1-sided alternative.

C3 =AVERAGE(A2:A61)

	A	B	C	D	E
1	Score	n	60		
2	76	mu0	90		
3	85	xbar	74.88333333		
4	78	sd	10.61673075		
5	70	t_obs	-11.02911991		
6	78	P-Value	5.76993E-16		
7	75				
8	73				
9	70				

Figure 1a: Computing xbar in Excel

C4 =STDEV(A2:A61)

	A	B	C	D	E
1	Score	n	60		
2	76	mu0	90		
3	85	xbar	74.88333333		
4	78	sd	10.61673075		
5	70	t_obs	-11.02911991		
6	78	P-Value	5.76993E-16		
7	75				

Figure 1a: Computing sd in Excel

C5 =(C3-C2)/(C4/SQRT(C1))

	A	B	C	D	E	F
1	Score	n	60			
2	76	mu0	90			
3	85	xbar	74.88333333			
4	78	sd	10.61673075			
5	70	t_obs	-11.02911991			
6	78	P-Value	5.76993E-16			
7	75					

Figure 1c: Computing t_{OBS} in Excel

TDIST =TDIST(ABS(C5),C1-1,2)

TDIST(x, deg_freedom, tail

	A	B	C
1	Score	n	60
2	76	mu0	90
3	85	xbar	74.88333333
4	78	sd	10.61673075
5	70	t_obs	-11.02911991
6	78	P-Value	=TDIST(ABS(C
7	75		

Figure 1d: Computing P-value in Excel

Computing the confidence interval for mean μ

The 95% confidence interval for the mean μ of an approximately normal population from a sample of size n is given by:

$$L = \bar{x} - t_{n-1,.025}\frac{s}{\sqrt{n}}$$

$$U = \bar{x} + t_{n-1,.025}\frac{s}{\sqrt{n}}$$

$\bar{x}$ = sample mean or average, s = sample standard deviation

$t_{n-1,.025}$ = Upper 2.5% point from t-table with df n-1

The SCORE data is in cells A2:A61, and the sample size n was entered in Cell C1. We have also calculated the sample average (in Cell C3) and sample sd (in Cell C4). The lower endpoint of 95% confidence interval is calculated in Cell G4 (see Figure 1e) by typing the formula

=D6-TINV(0.05,D4-1)



Figure 1e: Calculating lower limit L

The upper limit U is similarly calculated by typing the formula

=D6+TINV(0.05,D4-1) in Cell G5.

Example 2 (2 Independent Samples T-Test): Measured weights 0f 20 '3 lbs hamburger meat' packets from Grocery store A and 15 from Grocery Store B are given in the data file 'Weights with 2 Independent Sample T-Test Calculations.xlsx'. Test to see if the true means of '3 lbs hamburger meat' packets from Grocery store A and Grocery Store B are equal.

The null hypothesis H_0: $\mu_1 = \mu_2$ is to be tested vs. the alternative H_1: $\mu_1 \neq \mu_2$.

Running the 2-sample t-test using the Excel function TTEST

Start with Grocery Store A data in cells A2:A21 and Grocery Store B data in cells B2:A16.
Go to Cell E2, and click on Formulas/More Functions/Statistical/TTEST, select range A2:A21 as Array1, and B2:B16 as Array2, Tails = 2 (for 2-sided alternative), and Type = 2 for running the t-test for Equal Variances Case, and click OK (see Figure 2a), which will return P-value of 0.063191.



Figure 2a: Running 2-sample T-Test for Equal Variance case

The T-Test for Unequal Variances Case is run the same way, the only difference is that Type = 3 in this case. The P-values for the 2-sample T-Test run both ways are shown below:

	P-Value
Equal Variances Assumed	0.063191
Unequal Variances	0.116863

Since the P-values in both cases > .05, the null hypothesis of equal means is not rejected for data of Example 2.

Note that Excel only outputs the P-values; to get estimates and intermediate results, use the method shown below.

Computing the confidence interval for difference in two means $\mu_1 - \mu_2$

Excel does not provide a function for calculating the confidence interval for difference in two means, so we have to calculate it using the following formulas:

EQUAL VARIANCES - Use Pooled sample Variance

$$\hat{\mu}_1 - \hat{\mu}_2 = \bar{x}_1 - \bar{x}_2$$

$$sd(\bar{x}_1 - \bar{x}_2) = s_{pooled}\sqrt{\frac{1}{n_1} + \frac{1}{n_2}}$$

$$\text{where } s_{pooled} = \sqrt{\frac{(n_1 - 1)s_1^2 + (n_2 - 1)s_2^2}{(n_1 + n_2 - 2)}}$$

95% Confidence Interval for $\mu_1 - \mu_2$:

$$\bar{x}_1 - \bar{x}_2 \pm t_{df,.05} \times s_{pooled}\sqrt{\frac{1}{n_1} + \frac{1}{n_2}}, \text{ where } df = n_1 + n_2 - 2$$

UNEQUAL VARIANCES - Satterthwaites's Approximate Formula

$$\hat{\mu}_1 - \hat{\mu}_2 = \bar{x}_1 - \bar{x}_2$$

$$sd(\bar{x}_1 - \bar{x}_2) = \sqrt{\frac{s_1^2}{n_1} + \frac{s_2^2}{n_2}}$$

$$df = \frac{\left[\frac{s_1^2}{n_1} + \frac{s_2^2}{n_2}\right]^2}{\frac{\left(s_1^2/n_1\right)^2}{n_1 - 1} + \frac{\left(s_2^2/n_2\right)^2}{n_2 - 1}}$$

Satterthwaite's Approximate Confidence Interval:

$$\bar{x}_1 - \bar{x}_2 \pm t_{df,.05} \times \sqrt{\frac{s_1^2}{n_1} + \frac{s_2^2}{n_2}}$$

$t_{df,.05}$=2-sided T-table value corresponding to probability .05

All of the formulas for running the 2-Sample T-Test (Equal Variance Case and Unequal Variance Case) and computing 95% Confidence Intervals are shown in Figure 2b and the results in Figure 2c.



	A	B	C	D	E	F
1	A	B		N	MEAN	s
2	3	2.65	A	20	=AVERAGE(A2:A21)	=STDEV(A2:A21)
3	2.95	2.83	B	15	=AVERAGE(B2:B16)	=STDEV(B2:B16)
4	3	3.1				
5	2.94	2.81				
6	3	3.67	ASSUMING EQUAL VARIANCES			
7	3.02	2.93	xbar1-xbar2	=E2-E3		
8	2.95	2.92	S2+pooled	=((D2-1)*F2^2+(D3-1)*F3^2)/(D2+D3-2)		
9	2.95	3.15	S_pooled	=SQRT(D8)		
10	2.95	3.04	s_pooled X sqrt(1/n1 + 1/n2)	=D9*SQRT((1/D2)+(1/D3))		
11	3.05	2.87	df	=D2+D3-2		
12	3.05	3.3	t(df,.05)	=TINV(0.05,D11)		
13	2.96	3.45	L	=D7-D12*D10		
14	2.94	3.53	U	=D7+D12*D10		
15	2.95	3.51	t_obs	=D7/D10		
16	3.07	3.01	P-value	=TDIST(ABS(D15),D11,2)		
17	3.03		REJECT H0			
18	2.94					
19	2.93					
20	2.96		UNEQUAL VARIANCES			
21	3.05		xbar1-xbar2	=E2-E3		
22			sd(xbar1-xbar2)	=SQRT(F2^2/D2 +F3^2/D3)		
23			num_df	=(F2^2/D2 + F3^2/D3)^2		
24			denom_df	=(F2^2/D2)^2/(D2-1) + (F3^2/D3)^2/(D3-1)		
25			df	=D23/D24		
26			t(df,.05)	=TINV(0.05,D25)		
27			L	=D21-D26*D22		
28			U	=D21+D26*D22		
29			t_obs	=D21/D22		
30			P	=TDIST(ABS(D29),D25,2)		

Figure 2b: Excel formulas for running t-test and calculating L and U of 95% confidence interval for $\mu_1 - \mu_2$



	A	B	C	D	E	F
1	A	B		N	MEAN	s
2	3	2.65	A	20	2.9845	0.046052
3	2.95	2.83	B	15	3.118	0.30746
4	3	3.1				
5	2.94	2.81				
6	3	3.67	**ASSUMING EQUAL VARIANCES**			
7	3.02	2.93	xbar1-xbar2	-0.1335		
8	2.95	2.92	S2+pooled	0.041325303		
9	2.95	3.15	S_pooled	0.203286259		
10	2.95	3.04	s_pooled X sqrt(1/n1 ·	0.069435476		
11	3.05	2.87	df	33		
12	3.05	3.3	t(df,.05)	2.034515287		
13	2.96	3.45	L	-0.274767538		
14	2.94	3.53	U	0.007767538		
15	2.95	3.51	t_obs	-1.922648296		
16	3.07	3.01	P-value	0.06319078		
17	3.03		REJECT H0			
18	2.94					
19	2.93					
20	2.96		**UNEQUAL VARIANCES**			
21	3.05		xbar1-xbar2	-0.1335		
22			sd(xbar1-xbar2)	0.080050826		
23			num_df	4.10642E-05		
24			denom_df	2.83748E-06		
25			df	14.47207444		
26			t(df,.05)	2.144786681		
27			L	-0.305191945		
28			U	0.038191945		
29			t_obs	-1.667690479		
30			P	0.117580684		

Figure 2c: Results of T-Test and 95% confidence interval for $\mu_1 - \mu_2$

Example 3 (Paired T-Test): The data file Burger_Sales.xlsx shows daily sales of two adjacent fast food places for 14 randomly selected days. Test to see if the average sales of the two fast food restaurants are equal.

The data in this example is PAIRED since the *sales for the two restaurants are for same day, and we will need to run the paired T Test for this example.* Start with data in cells A2:A15 (McB sales) and B2:B15 (DK sales), and click on Formulas/More Functions/Statistical/TTEST. Select input ranges, Tails = 2, Type = 1 (for Paired T-Test), click on OK (Figure 3), to get P-value of 0.033599. Since the P-value for Example 3 data is < .05, we conclude that the average sales at the two stores are not equal.



Figure 3: Running Paired T-Test for Example 3